## The symmetry breaking region

```
FSB[d_, R_] := Show[Plot[d - 2, {γ, d - R, d + R},
   PlotStyle → {Black, Dashed}, PlotRange → {{d - R, d + R}, {d - 2 - R, d - 2 + R}}],
  ListLinePlot[{{d, d - 2 - R}, {d, d - 2 + R}}, PlotStyle → {Black, Dashed}],
  Plot[d - 2 - (γ - d), {γ, d - R, d + R}, PlotStyle → Black], Plot[{(d - 2)/d γ, γ - 2},
   {γ, d - R, d + R}, PlotStyle → {{Thick, Black}, {Thick, Black}}, Filling → {1 → {2}},
   FillingStyle → GrayLevel[0.8], PlotRange → {{d - R, d + R}, {d - 2 - R, d - 2 + R}}],
  Plot[{-2 + d - Sqrt[4 + d^2 + γ^2 - 2 d (2 + γ)], -2 + d + Sqrt[4 + d^2 + γ^2 - 2 d (2 + γ)]},
   {γ, d - R, d + R}, PlotStyle → {{Thick, Black}, {Thick, Black}},
   PlotRange → {{d - R, d + R}, {d - 2 - R, d - 2 + R}}],
  Plot[{Min[(d - 2)/d γ, -2 + d - Sqrt[4 + d^2 + γ^2 - 2 d (2 + γ)]], (d - 2)/d γ}, {γ, d - R, 0},
   PlotStyle → {{Thick, Black}, {Thick, Black}}, Filling → {1 → {2}},
   FillingStyle → GrayLevel[0.6], PlotRange → {{d - R, d + R}, {d - 2 - R, d - 2 + R}}],
  Plot[{Max[(d - 2)/d γ, -2 + d + Sqrt[4 + d^2 + γ^2 - 2 d (2 + γ)]], (d - 2)/d γ}, {γ, 2 d, d + R},
   PlotStyle → {{Thick, Black}, {Thick, Black}}, Filling → {1 → {2}}, FillingStyle →
    GrayLevel[0.6], PlotRange → {{d - R, d + R}, {d - 2 - R, d - 2 + R}}], Ticks → None]
```

```
FSB[5, 10]
```



## The symmetry breaking region (local)

```
F1[d_] := Show[Plot[d - 2, {γ, -d / 2, d + 2}, PlotStyle → {Thick, Black, Dashed},
   PlotRange → {{-d / 2, d + 2}, {-d / 2, d}}], Plot[{(d - 2)/d γ, γ - 2}, {γ, -d / 2, d + 2},
   PlotStyle → {{Thick, Black}, {Thick, Black}}, Filling → {1 → {2}},
   FillingStyle → GrayLevel[0.8], PlotRange → {{-d / 2, d + 2}, {-d / 2, d}}], Ticks → None]

F3[d_] := Plot[{-2 + d - Sqrt[4 + d^2 + γ^2 - 2 d (2 + γ)], -2 + d + Sqrt[4 + d^2 + γ^2 - 2 d (2 + γ)]},
  {γ, -d / 2, d + 2}, PlotStyle → {{Thick, Black}, {Thick, Black}},
  PlotRange → {{-d / 2, d + 2}, {-d / 2, d}}, Ticks → None]

F3b[d_] := Plot[{-2 + d - Sqrt[4 + d^2 + γ^2 - 2 d (2 + γ)], (d - 2)/d γ}, {γ, -d / 2, 0},
  PlotStyle → {{Thick, Black}, {Thick, Black}}, Filling → {1 → {2}},
  FillingStyle → GrayLevel[0.6], PlotRange → {{-d / 2, d + 2}, {-d / 2, d}}, Ticks → None]
```

```
G[d_] := Show[F1[d], F3b[d], F3[d]]
```

```
G[3]
Show[%, PlotRange → {{-1, 1}, {-1.5, 1.5}}]
```





## The symmetry breaking region (local) and gap issues

```
F1[d_, p_] := Show[Plot[d - 2, {γ, -d / 2, d + 2},
    PlotStyle → {Thick, Black, Dashed}, PlotRange → {{-d / 2, d + 2}, {-d / 2, d}}],
  Plot[γ - 2, {γ, -d / 2, d + 2}, PlotStyle → {Thick, Black}], Plot[{(d - 2)/d γ, (-d - 2 p + d p + γ)/p},
   {γ, -d / 2, d + 2}, PlotStyle → {{Thick, Black}, {Thick, Black}}, Filling → {1 → {2}},
   FillingStyle → GrayLevel[0.8], PlotRange → {{-d / 2, d + 2}, {-d / 2, d}}], Ticks → None]
```

```
F2[d_, p_] := Plot[(d^2 (-1 + p^2) + 2 d (p (-2 + γ) + γ) - (p (-2 + γ) + γ)^2)/(2 p (1 + p) (d - γ)),
  {γ, -d / 2, d + 2}, PlotStyle → {Thick, Black, Dotted},
  PlotRange → {{-d / 2, d + 2}, {-d / 2, d}}, Ticks → None]
```

```
F3[d_, p_] := Plot[{-2 + d - Sqrt[4 + d^2 + γ^2 - 2 d (2 + γ)], -2 + d + Sqrt[4 + d^2 + γ^2 - 2 d (2 + γ)]},
  {γ, -d / 2, d + 2}, PlotStyle → {{Thick, Black}, {Thick, Black}},
  PlotRange → {{-d / 2, d + 2}, {-d / 2, d}}, Ticks → None]
```

```
F3b[d_, p_] := Plot[{Max[-2 + d - Sqrt[4 + d^2 + γ^2 - 2 d (2 + γ)], (-d - 2 p + d p + γ)/p], (d - 2)/d γ},
  {γ, -d / 2, 0}, PlotStyle → {{Thick, Black}, {Thick, Black}}, Filling → {1 → {2}},
  FillingStyle → GrayLevel[0.6], PlotRange → {{-d / 2, d + 2}, {-d / 2, d}}, Ticks → None]
```

```
G[d_, p_] := Show[F1[d, p], F3b[d, p], F2[d, p], F3[d, p]]
```

```
G[3, 2]
Show[%, PlotRange → {{-0.6, 0.6}, {-1.4, 0.4}}]
Show[%, PlotRange → {{-0.4, 0.6}, {-0.7, 0.2}}]
```







## The three cases for the spectrum

```
Fess[n_, η_, δmax_] := Plot[((n - 2)/2 - δ)^2, {δ, 0, δmax},
  PlotStyle → {Black, Thick}, AspectRatio → 1.2, Ticks -> None]
F1L[n_, η_, δmax_] := Plot[2 η δ, {δ, 0, η + (n - 2)/2},
  PlotStyle → {Black, Thick, Dotted}, AspectRatio → 1.2, Ticks -> None]
F1R[n_, η_, δmax_] := Plot[2 η δ, {δ, η + (n - 2)/2, δmax},
  PlotStyle → {Black, Thick}, AspectRatio → 1.2, Ticks -> None]
F2L[n_, η_, δmax_] := Plot[2 (2 δ - n), {δ, 0, (n + 2)/2},
  PlotStyle → {Black, Thick, Dotted}, AspectRatio → 1.2, Ticks -> None]
F2R[n_, η_, δmax_] := Plot[2 (2 δ - n), {δ, (n + 2)/2, δmax},
  PlotStyle → {Black, Thick}, AspectRatio → 1.2, Ticks -> None]
F[n_, η_, δmax_] := Show[Fess[n, η, δmax], F1L[n, η, δmax],
  F1R[n, η, δmax], F2L[n, η, δmax], F2R[n, η, δmax]]

F[3, 3.5, 10]
F[3, 1.4, 7]
F[3, 0.35, 4]
```







## Section 1: Introduction and main results

```
θ = (d - γ) (p - 1) / (p (d + β + 2 - 2 γ - p (d - β - 2)));
pstar = (d - γ)/(d - 2 - β);
m1 = m /. Solve[pstar == 1/(2 m - 1), m][[1]]
mc = m /. Solve[d - γ + (2 + β - γ)/(m - 1) == 0, m][[1]]
Solve[p == pstar, β][[1]]
Simplify[1/(1 - m1)]
```

$$\frac{-2+2\,d-\beta-\gamma}{2\,(d-\gamma)}$$

$$\frac{-2+d-\beta}{d-\gamma}$$

$$\left\{\beta\to\frac{-d-2\,p+d\,p+\gamma}{p}\right\}$$

$$\frac{2\,(d-\gamma)}{2+\beta-\gamma}$$

### *The restriction due to p*

```
Solve[(d - γ)/(2 p) == (d - 2 - β)/2 t + (d - γ)/(p + 1) (1 - t), t][[1]]
Simplify[θ - t /. %]
θ /. β → γ - 2
Simplify[θ /. β → γ - 2]
Simplify[θ /. β → (d - 2)/d γ]
Simplify[% /. p → d/(d - 2)]
```

$$\left\{t\to-\frac{(-1+p)\,(d-\gamma)}{p\,(-2-d-2\,p+d\,p-\beta-p\,\beta+2\,\gamma)}\right\}$$

$$0$$

$$\frac{(-1+p)\,(d-\gamma)}{p\,(d-p\,(d-\gamma)-\gamma)}$$

$$-\frac{1}{p}$$

$$\frac{d-d\,p}{p\,(d\,(-1+p)-2\,(1+p))}$$

$$1$$

```
Solve[-2 - d - 2 p + d p - β - p β + 2 γ == 0, β][[1]]
Simplify[β - (d - 2 - (d - γ)/p) /. %]
Simplify[% /. Solve[p == (d - γ)/(d - 2 - β), β][[1]]]
```

$$\left\{\beta \to \frac{-2 - d - 2\,p + d\,p + 2\,\gamma}{1 + p}\right\}$$

$$\frac{d - d\,p - \gamma + p\,\gamma}{p + p^2}$$

$$\frac{d - d\,p - \gamma + p\,\gamma}{p + p^2}$$

- ***Self-similar solutions and self-similar variables***

```
R1[t_] := (t/ρ)^ρ
Simplify[FullSimplify[PowerExpand[R1'[t] - R1[t]^(γ-β-(m-1) (d-γ)-1)]] /. ρ → 1/((m - mc) (d - γ))]
R2[t_] := (1 + (2 + β - γ)/ρ t)^ρ
Simplify[
 FullSimplify[PowerExpand[R2'[t] - (2 + β - γ) R2[t]^(γ-β-(m-1) (d-γ)-1)]] /. ρ → 1/((m - mc) (d - γ))]
```

0

0

- ***Statement of Theorem 3***

```
Resnalpha = Solve[{n == (d - β - 2)/α + 2, n == (d - γ)/α, δ == 1/(1 - m), p == 1/(2 m - 1)}, {n, α, δ, m}][[1]];
αFS = Sqrt[(d - 1)/(n - 1)];
β /. Solve[α == αFS /. Resnalpha, β]
```

$$\left\{-2 + d - \sqrt{4 - 4\,d + d^2 - 2\,d\,\gamma + \gamma^2}\,,\ -2 + d + \sqrt{4 - 4\,d + d^2 - 2\,d\,\gamma + \gamma^2}\right\}$$

```
Lambda10 = Simplify[2 α^2 (2 δ - n)];
Lambda01 = 2 α^2 δ η;
ResDeltaEta =
  Solve[{-1 + d + 2 α^2 η - n α^2 η - α^2 η^2 == 0, Lambda10 == Lambda01}, {η, δ}][[2]];
Simplify[δ /. ResDeltaEta];
Simplify[p /. Solve[{% == 1/(1 - m), p == 1/(2 m - 1)}, {m, p}][[1]]];
FullSimplify[% /. Resnalpha];
Simplify[β /. Solve[% == p, β][[1]]];
Simplify[% + ((d - γ + p (d + 2 - γ)) (d - γ - p (d - 2 + γ)))/(2 p (1 + p) (d - γ))]
Simplify[η /. ResDeltaEta];
{%, Simplify[% /. Resnalpha]}
Simplify[α^2 /. Solve[{-1 + d + 2 α^2 η - n α^2 η - α^2 η^2 == 0, Lambda10 == Lambda01}, {η, α}][[2]]]
```

$$0$$

$$\left\{\frac{2\,\alpha - n\,\alpha + \sqrt{-4 + 4\,d + (-2 + n)^2\,\alpha^2}}{2\,\alpha},\ \frac{2 - d + \beta + \sqrt{d^2 - 2\,d\,\beta + \beta\,(4 + \beta)}}{2 + \beta - \gamma}\right\}$$

$$-\frac{(-1 + d)\,\delta^2}{n\,(n - 2\,\delta)\,(-1 + \delta)}$$

```
Simplify[(1 - m) {Lambda01, Lambda10} /. Resnalpha]
```

$$\left\{\frac{1}{2}\,(2 + \beta - \gamma)^2\,\eta,\ \frac{(2 + \beta - \gamma)\,(d - d\,p + p\,(4 + 2\,\beta - \gamma) - \gamma)}{2\,p}\right\}$$

- ***Statement of Proposition 4***

```
Lambda10 = Simplify[2 α^2 (2 δ - n)];
LambdaEss = α^2 ((n - 2)/2 - δ)^2;
Solve[{Lambda10 == LambdaEss}, δ]
Solve[{Lambda01 == LambdaEss}, δ]
```

$$\left\{\left\{\delta \to \frac{2 + n}{2}\right\},\ \left\{\delta \to \frac{2 + n}{2}\right\}\right\}$$

$$\left\{\left\{\delta \to \frac{1}{2}\left(-2 + n + 2\,\eta - 2\,\sqrt{-2\,\eta + n\,\eta + \eta^2}\right)\right\},\ \left\{\delta \to \frac{1}{2}\left(-2 + n + 2\,\eta + 2\,\sqrt{-2\,\eta + n\,\eta + \eta^2}\right)\right\}\right\}$$

## Section 2: the variational point of view

```
Simplify[θ /. p → pstar]
```

$$1$$

```
Simplify[θ - Simplify[θ /. {β → 2 (d - 2) - β, γ → 2 d - γ}]]
```

$$0$$

- ***Felli & Schneider***

```
ac = (d - 2)/2;
bFS = (d (ac - a))/(2 Sqrt[(ac - a)^2 + d - 1]) - (ac - a);
Eqn = Simplify[(α^2 - αFS^2)/(2 + β - γ) /. Resnalpha];
Simplify[{γ == 2 pstar b, 2 a - β == 0} /. Resnalpha];
Simplify[Eqn /. Solve[%, {β, γ}]][[1]];
FullSimplify[Solve[0 == %, b]];
Simplify[b/bFS /. %][[2]]
Eqn
hFS = (γ - d)^2 - (β - d + 2)^2 - 4 (d - 1);
Simplify[(8 d - 4 (2 + β + γ)) Eqn - hFS]
```

1

$$\frac{2\ (-2+d)\ \beta-\beta^2+\gamma\ (-2\ d+\gamma)}{8\ d-4\ (2+\beta+\gamma)}$$

0

- ***Weighted Gagliardo-Nirenberg***

```
Simplify[θ/2 + (1 - θ)/(p + 1) - 1/(2 p)]
```

$$\frac{(-1+p)\ (2+\beta-\gamma)}{2\ p\ (2+d-d\ p+\beta+p\ (2+\beta)-2\ \gamma)}$$

- ***Linear stability analysis***

```
v[s_] := (1 + s^2)^(-1/(p-1))
AA = Integrate[v'[s]^2 s^(n-1), {s, 0, ∞}, Assumptions → p > 1 && p < n/(n - 2)];
BB = Integrate[v[s]^(p+1) s^(n-1), {s, 0, ∞}, Assumptions → p > 1 && p < n/(n - 2)];
CC = Integrate[v[s]^(2 p) s^(n-1), {s, 0, ∞}, Assumptions → p > 1 && p < n/(n - 2)];
```

```
HSimp = {Gamma[2 p/(-1 + p)] → (1 + p)/(-1 + p) Gamma[(1 + p)/(-1 + p)], Gamma[1 + n/2] → n/2 Gamma[n/2],
    Gamma[-n/2 + 2 p/(-1 + p)] → (1 - n/2 + 2/(-1 + p)) Gamma[1 - n/2 + 2/(-1 + p)],
    Gamma[(1 + p)/(-1 + p)] → 2/(-1 + p) Gamma[2/(-1 + p)]};
Simplify[FullSimplify[(p - 1)^2/4 AA - BB + CC] /. HSimp] /. HSimp
Simplify[FullSimplify[n (p - 1)/(2 (p + 1)) BB - BB + CC] /. HSimp]
Simplify[FullSimplify[AA/BB - 2 n/(p^2 - 1)] /. HSimp] /. HSimp
Simplify[FullSimplify[AA/CC] - 4 n/((p - 1) (n + 2 - p (n - 2))) /. HSimp]
```

0

0

0

0

```
Simplify[{p (1 - θ)/θ 2 n/(p^2 - 1), (2 p - 1)/θ 4 n/((p - 1) (n + 2 - p (n - 2)))} /. Resnalpha]
```

$$\left\{\frac{4\,p\,(d-d\,p+p\,(2+\beta)-\gamma)}{(-1+p)^2\,(2+\beta-\gamma)},\ \frac{4\,p\,(-1+2\,p)}{(-1+p)^2}\right\}$$

## ▪ Section 3: the flow

```
v[r_] := (1 + r^2)^(-1/(p-1))
(-1 + p)^2 r^(1-n) (1 + r^2)^(2 - 1/(1-p))
  FullSimplify[PowerExpand[-A α^2 D[r^(n-1) v'[r], r] + B r^(n-1) v[r]^p - C r^(n-1) v[r]^(2 p-1)]]
{% == 0, D[%, {r, 2}] == 0} /. r → 0
Simplify[Solve[%, {A, B}]][[1]]
Simplify[{B/A, C/A} /. %]
Simplify[{p %[[1]], (2 p - 1) %[[2]]} /. Resnalpha]
```

$$(-1+p)^2\left(B-C+B\,r^2\right)+2\,A\left(-2\,p\,r^2+n\,(-1+p)\left(1+r^2\right)\right)\alpha^2$$

$$\left\{(B-C)\,(-1+p)^2+2\,A\,n\,(-1+p)\,\alpha^2==0,\ 2\,B\,(-1+p)^2+2\,A\,(2\,n\,(-1+p)-4\,p)\,\alpha^2==0\right\}$$

$$\left\{A\to\frac{C\,(-1+p)^2}{4\,p\,\alpha^2},\ B\to\frac{C\,(n+2\,p-n\,p)}{2\,p}\right\}$$

$$\left\{\frac{2\,(n+2\,p-n\,p)\,\alpha^2}{(-1+p)^2},\ \frac{4\,p\,\alpha^2}{(-1+p)^2}\right\}$$

$$\left\{\frac{p\,(2+\beta-\gamma)\,(d-d\,p+p\,(2+\beta)-\gamma)}{(-1+p)^2},\ \frac{p\,(-1+2\,p)\,(2+\beta-\gamma)^2}{(-1+p)^2}\right\}$$

```
H[λ_] := A λ^-a + B λ^b
Simplify[Solve[{θ/2 ξ == b/(a + b), (1 - θ)/(p + 1) η == a/(a + b), a == n - 2 - n/p, b == n ((p + 1)/(2 p) - 1),
    θ == Theta, n == (d - β - 2)/α + 2, n == (d - γ)/α}, {a, b, ξ, η, Theta, β, γ}]][[1]];
{η,
  ξ/η} /.
 %
```

$$\left\{\frac{2\,p\,(n\,(-1+p)-2\,(1+p))}{n\,(-1+p)-4\,p},\ 1\right\}$$

## Section 4: the spectrum

## *Calcul de la valeur propre k=1, l=0*

```
Lambda10 = Simplify[2 α^2 (2 δ - n)]
F[r_] := r^2 - c
Simplify[r^(1-n) (1 + r^2)^(1+δ) Simplify[α^2 D[r^(n-1)/(1 + r^2)^δ F'[r], r] + λ r^(n-1)/(1 + r^2)^(δ+1) F[r]] /.
  {c -> -n/(n - 2 δ), λ -> Lambda10}]
```

$-2\,\alpha^2\,(n-2\,\delta)$

0

```
F[r_] := r^2 + n/(n - 2 δ)
Simplify[α^2 D[r^(n-1)/(1 + r^2)^δ F'[r], r] + 2 α^2 (2 δ - n) r^(n-1)/(1 + r^2)^(δ+1) F[r]]
```

0

## *Calcul de la valeur propre k=0, l=1*

```
Lambda01 = 2 α^2 δ η
F[r_] := r^η
Res = Simplify[r^(3-n-η) (1 + r^2)^(1+δ)
   Simplify[- α^2 D[r^(n-1)/(1 + r^2)^δ F'[r], r] + r^(n-3)/(1 + r^2)^δ (d - 1) F[r] - λ r^(n-1)/(1 + r^2)^(δ+1) F[r]]]
a = Res /. r -> 0
b = Res - % /. r -> 1
Simplify[a - b /. λ -> Lambda01]
```

$2\,\alpha^2\,\delta\,\eta$

$-1+d\,(1+r^2)+2\,\alpha^2\,\eta-n\,\alpha^2\,\eta-\alpha^2\,\eta^2-r^2\,(1+\alpha^2\,\eta\,(-2+n-2\,\delta+\eta)+\lambda)$

$-1+d+2\,\alpha^2\,\eta-n\,\alpha^2\,\eta-\alpha^2\,\eta^2$

$-1+d-\alpha^2\,\eta\,(-2+n-2\,\delta+\eta)-\lambda$

0

```
Simplify[b /. λ → Lambda01]
Simplify[Solve[% == 0, η]]
```

$-1 + d - \alpha^2 \eta (-2 + n + \eta)$

$$\left\{\left\{\eta \to -\frac{-2\alpha + n\alpha + \sqrt{-4 + 4d + (-2+n)^2 \alpha^2}}{2\alpha}\right\}, \left\{\eta \to \frac{2\alpha - n\alpha + \sqrt{-4 + 4d + (-2+n)^2 \alpha^2}}{2\alpha}\right\}\right\}$$

- ***Condition Lambda01 = Lambda10***

```
Simplify[Lambda01 - Lambda10]
```

$2\alpha^2 (n + \delta(-2 + \eta))$

$$\text{Simplify}\left[-1 + d + 2\alpha^2 \eta - n\alpha^2 \eta - \alpha^2 \eta^2 \;/.\; \eta \to 2 - \frac{n}{\delta}\right]$$

```
% /. α^2 → x
Solve[% == 0, x][[1]]
x - α^2 /. %
Simplify[% /. Resnalpha]
Simplify[Solve[% == 0, β]]
```

$$\frac{n^2 \alpha^2 (-1 + \delta) - 2 n \alpha^2 (-1 + \delta) \delta + (-1 + d) \delta^2}{\delta^2}$$

$$\frac{n^2 x (-1 + \delta) - 2 n x (-1 + \delta) \delta + (-1 + d) \delta^2}{\delta^2}$$

$$\left\{x \to -\frac{(-1 + d)\delta^2}{n (n - 2\delta)(-1 + \delta)}\right\}$$

$$-\alpha^2 - \frac{(-1 + d)\delta^2}{n (n - 2\delta)(-1 + \delta)}$$

$$\frac{1}{4}(2 + \beta - \gamma)^2 \left(-1 - \frac{4(-1 + d)p^2}{(1 + p)(d - \gamma)(d(-1 + p) + \gamma + p(-4 - 2\beta + \gamma))}\right)$$

$$\left\{\{\beta \to -2 + \gamma\}, \{\beta \to -2 + \gamma\}, \left\{\beta \to \frac{d^2 \left(-1 + p^2\right) + 2 d (p(-2 + \gamma) + \gamma) - (p(-2 + \gamma) + \gamma)^2}{2 p (1 + p)(d - \gamma)}\right\}\right\}$$

- ***Comparisons***

```
LambdaStar = 2 δ α^2
Simplify[Lambda01 - LambdaEss]
Simplify[Lambda10 - LambdaEss]
Simplify[Lambda01 - LambdaStar]
Simplify[Lambda10 - LambdaStar]
```

$2\alpha^2 \delta$

$$\alpha^2 \left(-\frac{1}{4}(n - 2(1 + \delta))^2 + 2\delta\eta\right)$$

$$-\frac{1}{4}\alpha^2 (2 + n - 2\delta)^2$$

$2\alpha^2 \delta (-1 + \eta)$

$2\alpha^2 (-n + \delta)$

- ***Integrability of the eigenfunction (case k=0, l=1)***

```
Simplify[η + n/2 - (2 - m)/(1 - m) /. ResDeltaEta]
Simplify[δ /. Solve[{% == 0, δ == 1/(1 - m)}, {δ, m}][[1]]]
```

$$\frac{2\,\alpha + (-1+m)\sqrt{-4+4\,d+(-2+n)^2\,\alpha^2}}{2\,(-1+m)\,\alpha}$$

$$\frac{\sqrt{-4+4\,d+(-2+n)^2\,\alpha^2}}{2\,\alpha}$$

- ***Corollary 10***

```
Simplify[α^2 {δ (δ + 2 - n), δ (δ + 2)} /. Resnalpha];
Simplify[% - p (2 + β - γ) (d - γ - p (d - 2 - β))/(-1 + p)^2 /. Solve[δ == 2 p/(-1 + p), p][[1]]];
{%[[1]] /. (2 + β - γ) → 2 α, %[[2]]}
```

$$\left\{2\,\alpha^2\,\delta,\ \frac{1}{2}\,(2+d+\beta-2\,\gamma)\,(2+\beta-\gamma)\,\delta\right\}$$

- ***Appendix B***

```
Simplify[α^2 η (η + n - 2) /. η → (β + 1)/α];
Simplify[% /. Resnalpha]
```

$$(-1+d)\,(1+\beta)$$

```
Solve[LambdaEss == 0, δ]
δ1 = δ /. %[[2]]
Solve[Lambda10 == LambdaEss, δ]
δ2 = δ /. %[[2]]
```

$$\left\{\left\{\delta\to\frac{1}{2}\,(-2+n)\right\},\ \left\{\delta\to\frac{1}{2}\,(-2+n)\right\}\right\}$$

$$\frac{1}{2}\,(-2+n)$$

$$\left\{\left\{\delta\to\frac{2+n}{2}\right\},\ \left\{\delta\to\frac{2+n}{2}\right\}\right\}$$

$$\frac{2+n}{2}$$

```
δ3 = Simplify[η + (n - 2)/2 /. ResDeltaEta]
Simplify[% /. Resnalpha]
```

$$\frac{\sqrt{-4+4\,d+(-2+n)^2\,\alpha^2}}{2\,\alpha}$$

$$\frac{\sqrt{d^2-2\,d\,\beta+\beta\,(4+\beta)}}{2+\beta-\gamma}$$

```
Simplify[δ2 - δ3 /. ResDeltaEta]
```

$$\frac{1}{2}\left(2+n-\frac{\sqrt{-4+4\,d+(-2+n)^2\,\alpha^2}}{\alpha}\right)$$

```
{Lambda01, LambdaEss} / α^2;
Simplify[δ /. Solve[%[[1]] == %[[2]], δ][[1]]];
δ4 = % /. Sqrt[η (-2 + n + η)] → Sqrt[d - 1]/α
Simplify[% - (n + 2)/2 /. ResDeltaEta];
Simplify[Lambda01 - Lambda10 /. Solve[% == 0, η]]
```

$$-1 + \frac{n}{2} - \frac{\sqrt{-1 + d}}{\alpha} + \eta$$

$$\left\{2\,\alpha^2\,(n + \delta\,(-2 + \eta))\right\}$$

```
{Lambda01, Lambda10} / α^2
Simplify[δ /. Solve[%[[1]] == %[[2]], δ][[1]]]
FullSimplify[% /. ResDeltaEta]
Simplify[% /. Resnalpha]
```

$$\{2\,\delta\,\eta,\ -2\,(n - 2\,\delta)\}$$

$$-\frac{n}{-2 + \eta}$$

$$\frac{2\,n\,\alpha}{(2 + n)\,\alpha - \sqrt{-4 + 4\,d + (-2 + n)^2\,\alpha^2}}$$

$$\frac{2\,(d - \gamma)}{2 + d + \beta - \sqrt{d^2 - 2\,d\,\beta + \beta\,(4 + \beta)} - 2\,\gamma}$$

```
δ5 = Simplify[δ /. Solve[Lambda01 == Lambda10, δ]][[1]]
FullSimplify[% /. ResDeltaEta]
FullSimplify[% /. Resnalpha]
```

$$-\frac{n}{-2 + \eta}$$

$$\frac{2\,n\,\alpha}{(2 + n)\,\alpha - \sqrt{-4 + 4\,d + (-2 + n)^2\,\alpha^2}}$$

$$\frac{2\,(d - \gamma)}{2 + d + \beta - \sqrt{(d - \beta)^2 + 4\,\beta} - 2\,\gamma}$$

```
FullSimplify[δ5 - n]
FullSimplify[δ5 - δ2]
Simplify[2 α^2 η (η + n - 2) - (d - 1) /. η → 4/(n + 2)]
Solve[% == 0, α]
```

$$-\frac{n\,(-1 + \eta)}{-2 + \eta}$$

$$\frac{4 - (2 + n)\,\eta}{2\,(-2 + \eta)}$$

$$1 - d + \frac{8\,n^2\,\alpha^2}{(2 + n)^2}$$

$$\left\{\left\{\alpha \to -\frac{\sqrt{-1 + d}\,(2 + n)}{2\,\sqrt{2}\,n}\right\},\ \left\{\alpha \to \frac{\sqrt{-1 + d}\,(2 + n)}{2\,\sqrt{2}\,n}\right\}\right\}$$